
Sommario—La gestione dell’assegnazione dei docenti di riferi-
mento o garanti in ambito universitario è un compito complesso
che coinvolge vari fattori organizzativi e un gran numero di dati.
L’automazione di questo processo può semplificare notevolmente
il lavoro amministrativo e migliorare l’efficienza operativa delle
università. È stato sviluppato un sistema basato su Answer Set
Programming (ASP), un paradigma logico adatto alla risoluzio-
ne di problemi combinatori complessi. Utilizzando Python e
Clingo, un potente solver ASP, è stata progettata una soluzione
capace di identificare automaticamente i docenti di riferimento,
a partire da un insieme di dati specifici. Il sistema sviluppato è
descritto nei suoi dettagli tecnici, con un focus sulla modellazione
logica, l’implementazione e l’analisi della complessità computa-
zionale. I risultati ottenuti sono discussi insieme alle potenzialità
di estensione del sistema.

I. INTRODUZIONE

L’identificazione dei docenti di riferimento o garanti per
uno specifico corso di studi rappresenta un processo che
annualmente coinvolge le università italiane. Tale compito
può risultare complesso, specialmente in presenza di strutture
organizzative articolate e di grandi volumi di dati. La necessità
di automatizzare e ottimizzare questa ricerca è quindi crucia-
le per migliorare l’efficienza delle attività amministrative e
accademiche.

In questo progetto, il problema è stato affrontato utilizzando
Answer Set Programming (ASP), un paradigma di program-
mazione logica dichiarativa particolarmente adatto alla risolu-
zione di problemi combinatori complessi. L’implementazione
è stata realizzata con l’ausilio di Python [1] e Clingo
[2], un solver open source ASP che combina il modello
di programmazione logica con strumenti di ottimizzazione
efficienti.

L’obiettivo principale è stato quello di progettare e imple-
mentare un sistema che, a partire da un insieme di dati resi
disponibili dagli uffici di competenza, consenta di individuare
in maniera automatica i docenti di riferimento o garanti in
base a criteri specifici. La relazione descrive le fasi del
lavoro, dall’analisi dei requisiti del problema alla modellazione
logica, presentando i dettagli dell’implementazione concreta
e un’analisi sulla sua complessità computazionale. Infine,
vengono discussi i risultati ottenuti e le possibili estensioni
del progetto.

II. BACKGROUND

A. Motivazioni

Attualmente, l’Università di Parma gestisce l’assegnazione
dei docenti in modo manuale, affrontando il processo in manie-
ra incrementale per ciascun corso di laurea. Il personale incari-
cato impiega settimane per ottenere una versione soddisfacente
della distribuzione, basandosi frequentemente su preferenze
informali e criteri non documentati. Questo approccio risulta
poco flessibile e difficilmente adattabile a nuove esigenze.
Inoltre, presenta significative limitazioni, come la difficoltà
nel gestire situazioni complesse e l’incapacità di ottimizzare
il processo in tempo reale, rendendo il sistema poco efficiente
e reattivo ai cambiamenti.

B. Answer Set Programming

L’Answer Set Programming (ASP) è un paradigma di
programmazione logica dichiarativa, particolarmente adatto
per risolvere problemi complessi di natura combinatoria che
richiedono soluzioni flessibili e ottimizzate. A differenza della
programmazione imperativa tradizionale, ASP si concentra
sulla descrizione del cosa deve essere risolto piuttosto che su
come farlo, utilizzando una forma di logica che rappresenta le
conoscenze del problema e le sue restrizioni (vincoli).

ASP si basa sulla teoria degli answer sets, ossia un in-
sieme di atomi letterali consistenti con le regole e i fatti
che costituiscono il programma. Il compito del solver ASP
è quello di trovare gli insiemi di valori che risolvono il
sistema di equazioni logiche, fornendo così soluzioni ottimali
o ammissibili.

L’uso di questo paradigma di programmazione è particolar-
mente indicato in ambiti in cui sono presenti vincoli complessi,
preferenze multiple e soluzioni che devono rispettare deter-
minati criteri. ASP permette di modellare in modo naturale
problemi che coinvolgono l’ottimizzazione, la pianificazione, e
la ricerca di soluzioni in scenari combinatori, in cui le variabili
e le relazioni tra esse sono numerose e intricate.

C. Il solver

Clingo [2] è un solver open-source per ASP, sviluppato
dal gruppo Potassco. È uno degli strumenti più potenti e diffusi
per risolvere problemi complessi di ottimizzazione e combina-
zione, combinando un motore di inferenza logica con capacità
avanzate di ottimizzazione. Nel nostro progetto, è stato integra-
to direttamente in Python utilizzando le API Python ufficiali

[3], le quali consentono di interagire facilmente con il solver
all’interno di ambienti Python. Questa integrazione permette
di automatizzare il processo di invocazione e gestione delle
soluzioni, facilitando l’elaborazione dei dati e l’ottimizzazione
delle assegnazioni in tempo reale.

D. Provenienza e contenuto dei dati di input

I dati di input sono stati forniti dall’U.O. Progettazione Di-
dattica e Assicurazione della Qualità [4], in collaborazione con
il prof. A. Dal Palù dell’Università di Parma. Questi includono
un insieme di tabelle e documenti eterogenei, comprendenti le
coperture dei corsi per l’anno accademico corrente, l’elenco
del personale docente e le informazioni relative alle immatri-
colazioni nei corsi di laurea. L’elaborazione è stata effettuata
utilizzando Python e si è rivelata particolarmente complessa
a causa dell’assenza di una sorgente dati unica e centralizzata.

III. MODELLAZIONE DEL PROBLEMA

A. Strutturazione della soluzione

Inizio

Elaborazione dati (Python)

Generazione fatti ground (Python)

Solver Clingo (Python API + ASP)

Generazione tabella garanti (Python)

Fine

Figura 1. Pipeline del processo di assegnazione dei garanti.

In Fig. 1 è illustrata la pipeline di lavoro. Prima dell’ela-
borazione automatica dei dati, è stato necessario effettuarne
una normalizzazione manuale. Questo passaggio ha permesso
di uniformare i formati e risolvere eventuali incongruenze nei
dati forniti, facilitando così le fasi successive. La generazione
dei fatti ground costituisce la base di conoscenza positiva,
utilizzata dal programma ASP principale per calcolare gli
answer set. Successivamente, il solver Clingo viene eseguito
tramite Python e i risultati ottenuti vengono convertiti in
formato tabellare per facilitarne la visualizzazione.

B. Implementazione

L’implementazione è stata progettata separando la logi-
ca principale per la generazione dei garanti accademici da
quella relativa alle preferenze specifiche. Tra le preferenze
considerate figurano:

• assegnazione automatica del ruolo di garante al presidente
del corso;

• massimizzazione del numero di docenti con contratto a
tempo indeterminato;

• minimizzazione del numero di ricercatori e docenti a
contratto coinvolti.

Nel Cod. 1 viene illustrato il codice di main.lp. Per
questioni di ottimizzazione in fase di grounding si è preferito,
ove possibile, utilizzare gli aggregati evitando così l’utilizzo
della negazione esplicita.

1 % se garante, allora insegna almeno una materia in
quel corso

2 1{
3 insegnamento(I) : insegna(docente(Matricola

), insegnamento(I), corso(Corso))
4 } :-
5 garante(docente(Matricola), corso(Corso)).
6

7 scelta(docente(Matricola), corso(Corso)) :-
8 docente(Matricola),
9 insegna(docente(Matricola), insegnamento(_)

, corso(Corso)).
10

11 % possibli garanti
12 M{
13 garante(docente(Matricola), corso(Corso)) :
14 scelta(docente(Matricola), corso(Corso))
15 }N :-
16 min_garanti(M, corso(Corso)),
17 max_garanti(N, corso(Corso)),
18 corso(Corso).
19

20 % un docente e’ scelto al piu’ una volta
21 0{
22 garante(docente(Matricola), corso(Corso2))

:
23 garante(docente(Matricola), corso(Corso2)),
24 Corso1!=Corso2
25 }0 :-
26 garante(docente(Matricola), corso(Corso1)).
27

28 % almeno X garanti a tempo indeterminato
29 X{
30 garante(docente(Matricola), corso(Corso)) :
31 garante(docente(Matricola), corso(Corso)),
32 indeterminato(docente(Matricola))
33 }M :-
34 min_indeterminato(X, corso(Corso)),
35 max_garanti(M, corso(Corso)),
36 corso(Corso).
37

38 % al piu’ X garanti ricercatore
39 {
40 garante(docente(Matricola), corso(Corso)) :
41 garante(docente(Matricola), corso(Corso)),
42 ricercatore(docente(Matricola))
43 }X :-
44 max_ricercatori(X, corso(Corso)),
45 corso(Corso).
46

47 % al piu’ X garanti a contratto
48 {
49 garante(docente(Matricola), corso(Corso)) :
50 garante(docente(Matricola), corso(Corso)),
51 contratto(docente(Matricola))
52 }X :-
53 max_contratto(X, corso(Corso)),
54 corso(Corso).
55

56 R{
57 garante(docente(Matricola), corso(Corso)) :
58 garante(docente(Matricola), corso(Corso)),
59 afferisce(docente(Matricola), settore(

Settore)),

60 di_riferimento(settore(Settore), corso(
Corso))

61 } :-
62 corso(Corso),
63 min_riferimento(R, corso(Corso)).
64

65 #show garante/2.

Codice 1. Codice ASP del file main.lp.

Le preferenze (Cod. 2) sono state espresse tramite l’utilizzo
di weak constraint. In particolare, sono stati definiti quattro
livelli di priorità1 sui quali intervenire:

• priorità 4: massimizzare l’assegnazione dei presidenti di
corso come garanti;

• priorità 3: massimizzare il numero di corsi per i quali è
soddisfatto il predicato ideale/1, dove i garanti sono
esclusivamente docenti a contratto indeterminato;

• priorità 2: massimizzare l’impiego di docenti a contratto
indeterminato come garanti;

• priorità 1: minimizzare il numero di corsi per i quali è
soddisfatto il predicato non_ideale/1, i cui garanti
comprendono almeno un ricercatore e/o un docente a
contratto a tempo determinato.

1 #minimize{
2 -1@4,
3 Matricola :
4 garante(docente(Matricola), corso(

Corso)),
5 presidente(docente(Matricola),

corso(Corso))
6 }.
7

8 #minimize{
9 -1@2,

10 Matricola :
11 garante(docente(Matricola), corso(

Corso)),
12 indeterminato(docente(Matricola))
13 }.
14

15 ideale(corso(Corso)) :-
16 N=#count{
17 docente(Matricola) :
18 garante(docente(Matricola), corso(

Corso)),
19 indeterminato(docente(Matricola))
20 },
21 corso(Corso),
22 max_garanti(Max, corso(Corso)),
23 N=Max.
24

25 #minimize{
26 -1@3,
27 ideale(corso(Corso)) :
28 ideale(corso(Corso))
29 }.
30

31 non_ideale(corso(Corso)) :-
32 N=#count{
33 docente(Matricola) :
34 garante(docente(Matricola), corso(

Corso)),
35 indeterminato(docente(Matricola))
36 },
37 corso(Corso),
38 max_garanti(Max, corso(Corso)),
39 N<Max.

1Si ricorda che l’ordine di importanza è inverso rispetto al livello di priorità
indicato.

40

41 non_ideale(corso(Corso)) :-
42 N=#count{
43 docente(Matricola) :
44 garante(docente(Matricola), corso(

Corso)),
45 ricercatore(docente(Matricola))
46 },
47 corso(Corso),
48 max_garanti(Max, corso(Corso)),
49 N>0.
50

51 non_ideale(corso(Corso)) :-
52 N=#count{
53 docente(Matricola) :
54 garante(docente(Matricola), corso(

Corso)),
55 contratto(docente(Matricola))
56 },
57 corso(Corso),
58 max_garanti(Max, corso(Corso)),
59 N>0.
60

61 #minimize{
62 1@1,
63 non_ideale(corso(Corso)) :
64 non_ideale(corso(Corso))
65 }.

Codice 2. Codice ASP del file preferenze.lp.

C. Spazio delle soluzioni

Si analizza ora la complessità computazionale in termini di
possibili configurazioni generate candidate ad essere soluzioni.

1 scelta(docente(Matricola), corso(Corso)) :-
2 docente(Matricola),
3 insegna(docente(Matricola), insegnamento(_)

, corso(Corso)).
4

5 % possibili garanti
6 M{
7 garante(docente(Matricola), corso(Corso)) :
8 scelta(docente(Matricola), corso(Corso))
9 }N :-

10 min_garanti(M, corso(Corso)),
11 max_garanti(N, corso(Corso)),
12 corso(Corso),
13 afferisce(corso(Corso), categoria_corso(

Categoria)).

Codice 3. Frammento del file main.lp.

Nel Cod. 3 è illustrata la regola che controlla i garanti
generati. Per ciascun corso di studi (95 in totale), esiste una
sola versione ground che rende veri i predicati presenti nel
corpo della regola; questo significa che il corpo verrà attivato
esattamente una volta. Inoltre M e N, rispettivamente minimo
e massimo richiesti, coincidono; quindi l’aggregato presente
in testa itera sulle possibili scelte per un determinato corso,
costruendo insiemi di esattamente N elementi. Indicati con
∆ il numero medio delle possibili scelta ottenute da ogni
corso e con k il numero medio di garanti richiesti, si ottiene
che questa regola genera circa

|Corsi | ·
(
∆

k

)
= 95 · ∆!

k! · (∆− k)!

possibili answer set. In media risultano ∆ = 39 e k = 10,
ossia circa

95 · 39!

10! · 29!
= 95 · 39 · 38 · . . . · 30

10!
≈ 6 · 1010

risultati. Lo spazio delle alternative da considerare cresce
dunque fattorialmente rispetto al numero medio di insegnanti
per ciascun corso.

L’impiego di vincoli espressi tramite aggregati consente di
escludere una significativa quantità di answer set non validi.
Ad esempio, la seguente regola:

1 % un docente e’ scelto al piu’ una volta
2 0{
3 garante(docente(Matricola), corso(Corso2))

:
4 garante(docente(Matricola), corso(Corso2)),
5 Corso1!=Corso2
6 }0 :-
7 garante(docente(Matricola), corso(Corso1)).

Codice 4. Frammento del file main.lp.

esclude tutti gli answer set in cui un docente risulta selezionato
come garante per più di un corso.

Le regole riportate nel Cod. 5 stabiliscono invece limiti mas-
simi e/o minimi per le sottocategorie di docenza, differenziate
in base al tipo di contratto.

1 % almeno X garanti a tempo indeterminato
2 X{
3 garante(docente(Matricola), corso(Corso)) :
4 garante(docente(Matricola), corso(Corso)),
5 indeterminato(docente(Matricola))
6 }M :-
7 min_indeterminato(X, corso(Corso)),
8 max_garanti(M, corso(Corso)),
9 corso(Corso).

10

11 % al piu’ X garanti ricercatore
12 {
13 garante(docente(Matricola), corso(Corso)) :
14 garante(docente(Matricola), corso(Corso)),
15 ricercatore(docente(Matricola))
16 }X :-
17 max_ricercatori(X, corso(Corso)),
18 corso(Corso).
19

20 % al piu’ X garanti a contratto
21 {
22 garante(docente(Matricola), corso(Corso)) :
23 garante(docente(Matricola), corso(Corso)),
24 contratto(docente(Matricola))
25 }X :-
26 max_contratto(X, corso(Corso)),
27 corso(Corso).

Codice 5. Frammento del file main.lp.

IV. RISULTATI

A. Punti chiave

In questa sezione vengono esposti i principali indicatori di
qualità relativi ai risultati ottenuti in varie casistiche.

Un primo fattore caratterizzante è emerso valutando le 4
preferenze espresse tramite i weak constraint. La generazione
di Answer Set, tenuto conto di tutti i vincoli, risulta molto
esosa e il programma non è in grado di fornire il risultato
ottimo in tempo utile. Nonostante non sia possibile garantire
la produzione della soluzione migliore a causa dei tempi
di esecuzione, le soluzioni fornite dal programma sono da
considerare corrette ed esaustive, in termini di configurazioni
dei docenti. A questo proposito è stato introdotto un limite di
tempo di 5 secondi per fermare il risolutore e procedere con
l’analisi dell’output.

B. Preferenze

È possibile ottenere risultati ottimali in tempi brevi (qualche
secondo) eliminando alcune preferenze, per facilitare il proces-
so di ottimizzazione. In particolare, si raggiungono prestazioni
elevate se vengono tenuti in considerazione solo i vincoli
relativi ai presidenti e ai professori a tempo indeterminato.
Gli esempi riportati in questa sezione sono stati svolti secondo
questa logica.

C. Input ridotto

I primi test sono stati eseguiti ponendo come obiettivo la
generazione di configurazioni solo per corsi di studio singoli o
a coppie, per verificare rapidamente la correttezza dei risultati.
Per fare un esempio pratico, il Corso di Laurea Triennale in
Informatica (codice 3027) offre 26 insegnamenti, distribuiti
su 20 docenti. La categoria del corso e le numerosità di
immatricolazioni richiedono la presenza di almeno 9 garanti,
tra cui:

• non meno di 5 professori a tempo indeterminato;
• non più di 4 ricercatori;
• non più di 2 professori a contratto.

Tabella I
GARANTI GENERATI PER IL CORSO 3027.

Docente Matricola SSD 2015

Dal Palù A. 6625 INF/01
De Filippis C. 34499 MAT/05
Bonnici V. 34181 INF/01
Guardasoni C. 6801 MAT/08
Benini A. 26131 MAT/03
De Pietri R. 5536 FIS/02
Zaffanella E. 5602 INF/01
Bergenti F. 204741 INF/01
Bagnara R. 5145 INF/01

La soluzione ottimale, in questo caso, viene generata in
meno di 1 secondo. Nonostante l’esempio sia banale, i vin-
coli imposti sono stati rispettati e la soluzione è facilmente
consultabile da parte dell’utente.

Eccezioni di rilievo: Per il Corso di Laurea Triennale in
Chimica (codice 3024), il numero di insegnamenti (e di con-
seguenza il numero di docenti) è molto alto; in particolare, si
riscontra un elevato numero di professori con contratto a tempo
indeterminato. In queste casistiche, il programma finisce per
valutare una quantità elevatissima di modelli, potenzialmente
ottimi, tutti tra loro equivalenti. Per questo motivo il solver
potrebbe evolvere in una situazione di “stallo”, non riuscendo
a dimostrare di aver trovato un modello ottimale. Nel Cod. 6
è riportato l’output in questione, da cui è possibile notare
come la maggior parte del tempo di risoluzione sia impiegato
nella ricerca (senza successo) di un modello migliore. Il tempo
indicato nel campo Unsat indica, infatti, il tempo trascorso
tra l’istante in cui è stato trovato l’ultimo modello e la
terminazione [5]. Sono riportati, cambiati di segno, i valori di
ottimizzazione raggiunti, ovvero 1 (presidente) e 9 (professori
a tempo indeterminato), che rappresentano i valori migliori per
il corso di laurea in questione.

Models : 6
Optimum : yes
Optimization : -1 -9
Calls : 1
Time : 76.021s

(Solving: 76.01s
1st Model: 0.00s
Unsat: 76.00s)

CPU Time : 75.992s

Codice 6. Statistiche clingo per il corso 3024.

Anche per questo motivo, è stato introdotto l’utilizzo di un
timer all’interno del codice Python, della durata di 5 secondi.
Nei test eseguiti, questo limite si è rivelato sufficiente per
fornire soluzioni che rispettassero i vincoli prestabiliti.

D. Input completo

La discrepanza tra i due esempi riportati nei test sui corsi
singoli mette in mostra la complessità del modello, tipica dei
problemi affrontati tramite il paradigma di programmazione
Answer Set.

Un’ulteriore conferma si ottiene con l’esecuzione del pro-
gramma tenendo in considerazione tutti i corsi dell’ateneo. La
ricerca del modello ottimo si conclude, infatti, quasi in maniera
istantanea, contrariamente a quanto ci si possa aspettare.
L’ipotesi è che avere più corsi e professori a disposizione
renda alcune soluzioni meno ottimali in maniera naturale,
alleggerendo il carico di lavoro dell’ottimizzatore.

Models : 196
Optimum : yes
Optimization : -76 -705
Calls : 1
Time : 0.800s

(Solving: 0.65s
1st Model: 0.36s
Unsat: 0.00s)

CPU Time : 0.714s

Codice 7. Statistiche clingo per tutti i corsi.

Massimizzazione delle preferenze: Utilizzando tutti i vincoli
presenti nel file delle preferenze, si ottengono configurazioni
più specifiche, in base alle proprietà descritte in dettaglio alla
sezione III-B. Nel Cod. 8 viene mostrato un esempio, con
un limite di tempo impostato a 5 secondi. Si notano alcune
proprietà a riguardo:

• la risoluzione richiede in generale più tempo;
• i livelli di ottimizzazione raggiunti relativi a presidenti e

professori a tempo indeterminato non cambiano di molto
rispetto all’esempio mostrato nel Cod. 7;

• il numero di corsi per i quali risulta vero il predicato
ideale/1 (43 corsi) è leggermente inferiore alla sua
controparte non_ideale/1 (52 corsi).

TIME LIMIT : 1
Models : 238+

Optimum : unknown
Optimization : -68 -43 -702 52
Calls : 1
Time : 5.003s

(Solving: 4.82s
1st Model: 1.38s
Unsat: 0.00s)

CPU Time : 4.870s

Codice 8. Statistiche clingo per tutti i corsi con preferenze massime.

V. CONCLUSIONE

L’esperimento descritto ha dimostrato l’efficacia dell’ap-
proccio basato su Answer Set Programming per affrontare
il problema dell’individuazione automatica dei garanti acca-
demici. Grazie all’uso di Clingo e Python, è stato possibile
implementare un sistema in grado di analizzare un grande
volume di dati e di risolvere un problema complesso in tempi
computazionalmente accettabili.

I risultati ottenuti dimostrano come l’approccio adottato
abbia soddisfatto pienamente i criteri richiesti, garantendo
una selezione accurata e conforme alle specifiche definite.
La modellazione logica, combinata con la capacità di ASP
di gestire vincoli complessi e criteri di ottimizzazione, si è
dimostrata uno strumento indispensabile per affrontare questo
problema.

Nonostante il successo del sistema, permangono alcune aree
di miglioramento. Ad esempio, l’introduzione della possibilità
che un docente possa essere garante per due corsi di studio
con un peso proporzionale (0,5 per ciascuno) potrebbe rendere
il sistema più flessibile, ma al costo di un aumento del tempo
di risoluzione. Inoltre, è necessaria la creazione di una base
dati centralizzata, aggiornata e accessibile in tempo reale; ciò
richiederebbe una ristrutturazione del codice Python esistente
ma si avrebbe il vantaggio di operare su dati completi e aggre-
gati. Sarebbe altresì opportuno sviluppare un’interfaccia utente
per rendere il sistema più accessibile e facile da utilizzare per
il personale accademico.

In conclusione, il progetto rappresenta un primo passo
promettente verso l’automazione intelligente delle attività ac-
cademiche. L’adozione futura di soluzioni avanzate, quali una
maggiore integrazione con i sistemi di gestione universita-
ria e l’uso di tecnologie complementari, potrebbe ampliare
ulteriormente l’applicabilità e l’impatto di questo lavoro.

